

Elderly and Child Poverty Rates

Goal

Households with children and/or the elderly are usually at higher risk of poverty. The rising interest in analysing the incomes of these groups has increased the use of child and elderly poverty rates. (See, for example, the LIS key figures.)

Activity

Use data for Finland and the US in 2000. Calculate the Head Count Ratio and the Income Gap Ratio for the total population, the elderly, and for children.

Guidelines

- Prepare the data as you did in the previous exercise (drop observations with missing or zero *dpi*).
- All surveyed households and their members must be included in the estimates of the poverty line. After the (unique) poverty line has been calculated, only those households with either members under the age of 18 (for the child poverty figures) or over 64 (elderly figures) are included when computing the proportion of the population (subgroup) living in poverty.
- One way to consider a subgroup in your calculations is to change the weights. In this case, you can create two additional weights: one for households with children and one for households with elderly. These weights will be equal to 0 if there are no children/elderly in the household, but will be equal to the normal weight multiplied by the number of children/elderly (and not total number of household members!):

```
gen cwt = hweight * d27
```

```
gen ewt = hweight * (num6574+numge75)
```

- For this exercise, construct the indicator three times: once for each group you want to examine. For each indicator, you will use the corrected weight.
- For this estimation, the use of the `poverty` ado file is a bit more complex. In fact, it usually automatically recalculates the poverty line as half or two thirds of the median for the population under consideration, unless one manually fixes the poverty line at a certain (different) value. In order to do this, you must first create the poverty line in a macro:

```
local povline = 0.5*r(p50)
```

and then use the macro in the poverty command:

```
poverty ey [aw=wt], line(`povline') h igr
```

Program

```
di "*** INCOME DISTRIBUTION I - Exercise 11 ***"

global keepit "hweight d4 d27 num6574 numge75 dpi"

program define pov
  use $keepit using $data, clear

  drop if dpi==. | dpi==0
  gen ey=(dpi/(d4^0.5))

  _pctile ey [w=hweight*d4], p(50)
  local povline = r(r1)*.5

  di "Results for the total population"
  poverty ey [aw=hweight*d4], line(`povline') h igr
  di "Results for children"
  poverty ey [aw=hweight*d27], line(`povline') h igr
  di "Results for the elderly"
  poverty ey [aw= hweight*(num6574+numge75)], line(`povline') h igr
end

foreach file in $fi00h $us00h {
  global data "`file'"
  di "$data"
  pov
}
```

Results

	# Observations	Poverty line	HCR	IGR
Finland 2000				
Total	<i>10 421</i>	<i>48,166.15</i>	<i>5.43%</i>	<i>25.57%</i>
with children	<i>3 893</i>	<i>48,166.15</i>	<i>2.97%</i>	<i>44.70%</i>
with elderly	<i>1 837</i>	<i>48,166.15</i>	<i>8.47%</i>	<i>12.20%</i>
US 2000				
Total	<i>49 351</i>	<i>12,046.99</i>	<i>17.05%</i>	<i>33.45%</i>
with children	<i>18 030</i>	<i>12,046.99</i>	<i>21.93%</i>	<i>34.13%</i>
with elderly	<i>11 634</i>	<i>12,046.99</i>	<i>24.72%</i>	<i>29.60%</i>

Comments

- As already mentioned above, the poverty line remains the same whether one examines the entire population, children, or the elderly but the households included in calculating the percentage below the threshold change with each subgroup. Therefore, keep in mind that the sample sizes of these subgroup rates are based on fractions of the entire sample and treat these figures with care when interpreting your results.